

TATE RESOLUTIONS ON PRODUCTS OF PROJECTIVE SPACES: COHOMOLOGY AND DIRECT IMAGE COMPLEXES

DAVID EISENBUD, DANIEL ERMAN, AND FRANK-OLAF SCHREYER

ABSTRACT. We describe the Macaulay2 package `TateOnProducts` and its capabilities, which include computing cohomology tables of sheaves on products of projective spaces and the derived category `pushForward` of a sheaf under a morphism from a projective scheme to a projective space.

INTRODUCTION

The main functions implemented in this package are:

- (1) `cohomologyHashTable`, which computes all of the cohomology groups of a coherent sheaf on a product of projective spaces, within a specified range of multidegrees;
- (2) `beilinsonMonad`, which computes the Beilinson monad of a sheaf on a product of projective spaces; and
- (3) `directImageComplex`, which computes the derived push forward of a coherent sheaf with respect to any of the natural projection maps from a product of projective spaces.

The algorithms employed exploit the Koszul duality between polynomial rings and exterior algebras, thus turning questions of cohomology of sheaves into questions about free resolutions of modules over an exterior algebra. The core algorithm of the package is the computation of a finite part of the **Tate resolution** of a coherent sheaf, which is a doubly infinite, multigraded, free complex of free modules over an exterior algebra whose. The functions mentioned above are easy to compute once this is done.

AMS Subject Classification: Primary: 14F05 , Secondary: 13D02, 14Q99

Keywords: cohomology groups, Tate resolutions, Beilinson monads

The first and second authors are grateful to the National Science Foundation for partial support. This work is a contribution to Project I.6 of the third author within the SFB-TRR 195 "Symbolic Tools in Mathematics and their Application" of the German Research Foundation (DFG).

Tate resolutions were introduced for sheaves on \mathbb{P}^n in [EFS03]. The Macaulay2 package BGG implements an algorithm for computing Tate resolutions for such sheaves, and for using them to compute sheaf cohomology and Beilinson monads [M2-BGG]. Tate resolutions for products of projective spaces are introduced in [EES15]. For a product of two or more projective spaces, there is a major new difficulty, as each term of the Tate resolution is a free module of infinite rank. In this package we compute a part of this infinite object that we call a **corner complex**. Corner complexes play an essential role in [EES15], and they are sufficient for our applications.

This paper is organized as follows. In Section 1, we briefly review the BGG correspondence and Tate resolution. In Section 2, we describe the finite part of a Tate resolution that we can actually compute. In Section 3, we discuss the command `cohomologyTable`. In Section 4, we provide a brief discussion of Beilinson monads, and discuss the command `beilinsonMonad`. In Section 5, we discuss derived push forward algorithms. In the final Section 6, we introduce the corner complexes, which play a mayor role in the construction of the Tate resolutions and in many of our applications.

ACKNOWLEDGMENTS

We thank Mike Stillman for major help with aspects of this package.

1. THE BGG CORRESPONDENCE AND TATE RESOLUTIONS

1.1. Review of Tate Resolutions for Projective spaces. Fix a field k and an $(n + 1)$ -dimensional vector space W . We let $S = \text{Sym } W$ be the symmetric algebra, with generators in degree 1. We also set $V = W^*$ and let $E = \Lambda V$ be the exterior algebra on E , with generators in degree¹ 1. The BGG correspondence stems from two simple observations.

First, if A, B, C are finite-dimensional vector spaces over k , then there is a natural bijection between homomorphisms $A \otimes_k B \rightarrow C$ and homomorphisms $B \rightarrow C \otimes_k A^*$.

Second, if $M = \bigoplus_{i \geq i_0} M_i$ is a finitely generated graded S -module, then the module structure induces a sequence of map $W \otimes M_i \rightarrow M_{i+1}$. By the correspondence above, this yields a sequence of maps $M_i \rightarrow M_{i+1} \otimes V$, which induces a sequence of linear maps of graded free E -modules

$$b_{gg}(M) : \cdots \rightarrow M_i \otimes E(i) \rightarrow M_{i+1} \otimes E(i+1) \rightarrow \cdots$$

¹Our conventions in this paper are consistent with the conventions in the Macaulay2 packages BGG and TateOnProducts. See Remark 1.2 for a detailed comparison of how these conventions differ from those in [EFS03, EES15].

A formal computation confirms that the conditions of commutativity and associativity of the action of S on M exactly correspond to the condition that $\text{bgg}(M)$ is a complex; that is, consecutive maps compose to 0.

In exactly the same way, a graded E -module N gives rise to a linear complex of free S -modules

$$\text{bgg}(N) : \dots \rightarrow N_i \otimes S(i) \rightarrow N_{i+1} \otimes S(i+1) \rightarrow \dots$$

Tate resolutions, as defined below, are exact complexes, and our computation depends on the fact that such a complex is determined by any differential. However, in general, only a truncation of the complex $\text{bgg}(M)$ is exact. This corresponds to a truncation of M at a degree sufficiently high to have a linear resolution over S . The following useful criterion shows that such a truncation is necessary, and sufficient:

Theorem 1.1 (Reciprocity). [EFS03, Theorem 3.7 and Corollary 2.4]. *With notation as above:*

- (1) $\text{bgg}(M)$ is an injective resolution of the E -module N if and only if $\text{bgg}(N)$ is a free resolution of M .
- (2) These conditions are satisfied if and only if the Castelnuovo-Mumford regularity of M is 0 and M has no submodule of finite length.

In their 1978 paper, Bernstein-Gelfand-Gelfand [BGG78] used the BGG correspondence to identify the derived category $D^b(\mathbb{P}^n)$ of bounded complexes of coherent sheaves on \mathbb{P}^n with the stable category of E -modules. In the same volume [Be78] Beilinson described $D^b(\mathbb{P}^n)$ in terms of exceptional sequences. (See Section 4 for more on Beilinson's work.)

Tate resolutions, as introduced in [EFS03], connect the approaches of [BGG78] and [Be78] and provide the foundation for the package [M2-BGG]. Let $M = \oplus M_d$ be a graded S -module representing the coherent sheaf \mathcal{F} on \mathbb{P}^n . For any r , the truncation $M_{\geq r} = \oplus_{d \geq r} M_d$ represents the same sheaf. If $r \gg 0$, then Theorem 1.1 shows that the complex

$$\text{bgg}(M_{\geq r}) : M_r \otimes E(r) \rightarrow M_{r+1} \otimes E(r+1) \rightarrow \dots$$

is acyclic. By combining this injective resolution with a minimal free resolution of $P = \ker(M_r \otimes E(r) \rightarrow M_{r+1} \otimes E(r+1))$, we obtain an infinite exact complex

$$\mathbf{T}(\mathcal{F}) : \dots \rightarrow T^{r-2} \rightarrow T^{r-1} \rightarrow T^r \rightarrow T^{r+1} \rightarrow \dots$$

of free E -modules, which depends only on \mathcal{F} . This is called the **Tate resolution of \mathcal{F}** . Any finite part can be found by computing syzygies over the exterior algebra.

By [EFS03, Theorem 4.1], the Betti numbers of $\mathbf{T}(\mathcal{F})$ encode the ranks of the sheaf cohomology groups of \mathcal{F} via the following formula:

$$\mathbf{T}^d(\mathcal{F}) = T^d = \sum_{i=0}^n H^i(\mathbb{P}^n, \mathcal{F}(d-i)) \otimes E(d-i).$$

and thus the syzygy computation yields the dimensions of the vector spaces $H^i(\mathbb{P}^n, \mathcal{F}(d))$ in any bounded range. The Beilinson monad of \mathcal{F} can also be obtained from the Tate resolution—see Section 4.

1.2. Tate Resolutions for Products of Projective Spaces. We now consider a product of projective spaces $\mathbb{P} = \mathbb{P}^{n_1} \times \dots \times \mathbb{P}^{n_t}$. The Cox ring of \mathbb{P} is a \mathbb{Z}^t -graded polynomial ring $S = \text{Sym}(W_1 \oplus W_2 \oplus \dots \oplus W_t)$ where $\dim W_i = n_i + 1$ and where the elements of W_1 have degree $(1, 0, \dots, 0)$, the elements of W_2 have degree $(0, 1, 0, \dots, 0)$ and so on. As before, we let $V_i = W_i^*$ for all i , and define a \mathbb{Z}^t -graded exterior algebra $E = \Lambda(V_1 \oplus V_2 \oplus \dots \oplus V_t)$ where $\deg(W_i) = \deg(V_i)$. When comparing multidegrees in \mathbb{Z}^t , we always take the termwise partial order.

By a nearly identical computation as before, the BGG correspondence $M \mapsto bgg(M)$ sends a multigraded S -module to a linear, multigraded, free, t -fold multicomplex of E -modules. For instance, if $t = 2$, then $bgg(M_{\geq(i,j)})$ is the total complex of a double complex:

$$(1) \quad \begin{array}{ccccc} & \vdots & & \vdots & \\ & \uparrow & & \uparrow & \\ M_{i,j+1} \otimes E(i, j+1) & \longrightarrow & M_{i+1,j+1} \otimes E(i+1, j+1) & \longrightarrow & \dots \\ & \uparrow & & \uparrow & \\ M_{i,j} \otimes E(i, j) & \longrightarrow & M_{i+1,j} \otimes E(i+1, j) & \longrightarrow & \dots \end{array}$$

In general, $bgg(M)$ is the total complex of a t -fold complex.

Remark 1.2. We follow the conventions of the package [M2-Tate], which differs from the conventions of [EES15]. In particular, in the paper:

- The exterior algebra E is negatively graded.
- We use ω_E instead of E .
- Tate resolutions are cochain complexes instead of chain complexes.

The theory of Tate resolutions for \mathbb{P} is developed in [EES15]. If M is a graded S -module representing \mathcal{F} , then the Tate resolution $\mathbf{T}(\mathcal{F})$ is an exact complex of free E -modules with terms

$$(2) \quad \mathbf{T}^d(\mathcal{F}) = \sum_{a \in \mathbb{Z}^t} H^{d-|a|}(\mathbb{P}, \mathcal{F}(a)) \otimes E(a),$$

where $|a| = a_1 + \dots + a_n$ denotes the total degree. This shows that the Tate resolution encodes all the cohomology groups $H^j(\mathbb{P}, \mathcal{F}(a))$. However, (2) also shows that, when $t > 1$, each term $\mathbf{T}^d(\mathcal{F})$ of the Tate resolution will be an infinitely generated E -module. Computing the entire Tate resolution $\mathbf{T}^d(\mathcal{F})$ is thus infeasible. We can nevertheless effectively compute the subquotient complex of $\mathbf{T}^d(\mathcal{F})$ which consists of all terms in a finite range $low \leq a \leq high$ of multidegrees with $low, high \in \mathbb{Z}^t$ and this is sufficient for all of our applications.

2. TATE RESOLUTIONS

Each of the major applications in this package stems from the computation of a bounded part of the Tate resolution. Let M be a finitely generated \mathbb{Z}^t -graded S -module, and let \mathcal{F} be the corresponding sheaf on \mathbb{P} . Let $low \leq high \in \mathbb{Z}^t$ be multi-degrees defining an interval. The function

`tateResolution(M, low, high)`

computes a subquotient complex of $\mathbf{T}(\mathcal{F})$ that contains all summands generated in degrees $low \leq a \leq high$. To compute this, Macaulay2 first uses `coarseMultigradedRegularity(M)` to find some $b \gg 0$ such that the total complex of $b\text{gg}(M_{\geq b})$ is acyclic.² Then, it resolves the kernel of the first map in $b\text{gg}(M_{\geq b})$, continuing until it covers all of the degrees in the desired range.

For instance, if $t = 2$ then we would resolve the kernel of the map:

$$M_b \otimes E(b) \rightarrow \begin{array}{c} M_{b+(1,0)} \otimes E(b+(1,0)) \\ \oplus \\ M_{b+(0,1)} \otimes E(b+(0,1)) \end{array}$$

This yields a multigraded complex of finitely generated, free E -modules which is denoted $\text{Tail}_b(M)$ in [EES15, §1]. A key fact from [EES15, §1] is that $\text{Tail}_b(M)$ equals the subquotient complex of $\mathbf{T}(\mathcal{F})$ obtained by restricting to degrees $\leq b - (1, 1, \dots, 1)$.³ By increasing b and/or by computing enough terms of the free resolution $\text{Tail}_b(M)$, we can thus compute any portion of $\mathbf{T}(\mathcal{F})$.

For example, let $\mathbb{P} = \mathbb{P}^1 \times \mathbb{P}^2$ and let $M = S^1$ be a module representing $\mathcal{F} = \mathcal{O}_{\mathbb{P}^1 \times \mathbb{P}^2}$. The input

`T = tateResolution(M, {-3, -3}, {0, 0})`

computes a multi-graded complex of the form

²This is closely related to the question of computing the multigraded regularity of M . Finding an efficient algorithm for that computation remains an interesting question.

³This coincidence is essential to the computability of $\mathbf{T}(\mathcal{F})$ within the desired range. It relies on properties of corner complexes, briefly described in Section 6 and treated in detail in [EES15, §3].

$$\begin{array}{cccccccc}
& 1 & 2 & 6 & 14 & 29 & 55 & 97 \\
0 & \leftarrow E & \leftarrow E & \leftarrow E & \leftarrow E & \leftarrow E & \leftarrow E & \leftarrow E \\
-1 & 0 & 1 & 2 & 3 & 4 & 5 & 6
\end{array}$$

which includes all the free summands in the Tate resolution that have generators in degree $(-3, -3) \leq a \leq (0, 0)$. The numbers below the free modules are the homological degrees. The Betti table of this complex with respect to total degree, is

```

betTi T
      0 1 2 3 4 5 6
total: 1 2 6 14 29 55 97
0: 1 . . . . .
1: . 1 2 3 4 5 6
2: . 1 3 6 10 15 21
3: . . 1 5 15 35 70

```

3. COMPUTING COHOMOLOGY TABLES

To compactly encode the dimensions of the sheaf cohomology groups of a coherent sheaf \mathcal{F} , we introduce the **Euler polynomial**

$$\sum_{i \geq 0} (\dim H^i(\mathbb{P}, \mathcal{F})) h^i \in \mathbb{Z}[h].$$

For concision, we often write $H^i(\mathcal{F})$ in place of $H^i(\mathbb{P}, \mathcal{F})$. Let M be a finitely generated \mathbb{Z}^t -graded S -module, and let \mathcal{F} be the corresponding sheaf on \mathbb{P} . Let $low \leq high \in \mathbb{Z}^t$ be multi-degrees defining an interval. We can compute all the cohomology vector spaces of all the twists $\mathcal{F}(a)$ for $low \leq a \leq high$ with the function

```
eulerPolynomialTable(M, low, high)
```

The output of this function is a hash table consisting of the pairs

$$a \Rightarrow \sum_{i \geq 0} (\dim H^i(\mathbb{P}, \mathcal{F}(a))) h^i \quad \text{where } a \in \mathbb{Z}^t.$$

As in the case of a single projective space, these are computed as Betti numbers of the appropriate finite part of the Tate resolution.

For example, if $\mathbb{P} = \mathbb{P}^1 \times \mathbb{P}^2$ and $\mathcal{F} = \mathcal{O}_{\mathbb{P}^1 \times \mathbb{P}^2}$ is represented by S^1 , then:

```

(S,E) = productOfProjectiveSpaces{1,2};
low  = {-3,-3}; high = {3,3};
eT = eulerPolynomialTable(S^1, low, high);
eT#{2,-3}

```

gives output 3h2. This means that $H^2(\mathcal{F}(2, -3))$ has rank 3 and that $H^i(\mathcal{F}(2, -3)) = 0$ for $i \neq 2$.

In the case $t = 2$, we can display the hash table as a matrix using the function `cohomologyMatrix(M, low, high)`, where the upper right hand corner of the table corresponds to the multi-index *high*: Continuing with the previous example, we have:

`cohomologyMatrix(S^1, low, high)`

gives output

	20h	10h	0	10	20	30	40	
	12h	6h	0	6	12	18	24	
	6h	3h	0	3	6	9	12	
	2h	h	0	1	2	3	4	
	0	0	0	0	0	0	0	
	0	0	0	0	0	0	0	
	2h3	h3	0	h2	2h2	3h2	4h2	

Since $high = \{3, 3\}$, the index of the top row is 3 as is the index of the right-hand column. The (i, j) -th entry is the Euler polynomial of $\mathcal{F}(i, j)$.

4. THE BEILINSON MONAD

Let \mathcal{F} be a coherent sheaf on \mathbb{P} . The Beilinson monad of \mathcal{F} is a complex of sheaves \mathcal{B} whose terms are direct sums of sheaves from an exceptional collection on \mathbb{P} built from exterior powers of the cotangent bundles on the factors, and whose homology groups satisfy

$$H^i \mathcal{B} = \begin{cases} \mathcal{F} & \text{if } i = 0 \\ 0 & \text{if } i \neq 0 \end{cases}.$$

The advantage of this representation is that the terms of the \mathcal{B} depend only on the cohomology groups of \mathcal{F} . See [Be78, AO89, EFS03, Huy06] for discussion of Beilinson monads and [EES15, §2] for a discussion in the context of \mathbb{P} . As explained in [EES15, §2], the Beilinson monad for \mathcal{F} is determined by the subquotient complex of $\mathbf{T}(\mathcal{F})$ consisting of summands generated in the finite range of degrees $0 \leq a \leq n$. This range of degrees is known as the *Beilinson window*, and a Beilinson monad for a \mathcal{F} can be computed easily from any finite part of the Tate resolution that contains the Beilinson window.

For the computation, we start with a module M representing a coherent sheaf \mathcal{F} on \mathbb{P} . Executing

`B = beilinsonMonad M`

produces a complex B of graded S -modules such that the corresponding complex \tilde{B} of sheaves on \mathbb{P} is the Beilinson monad for \mathcal{F} .

For example, if we take M to be a twist of the syzygy module of the maximal ideal in the bihomogeneous coordinate ring S of $\mathbb{P} = \mathbb{P}^1 \times \mathbb{P}^2$, we can execute:

```
M = S^{{1,1}} ** ker vars S;
```

The command `beilinsonWindow` computes the subquotient complex of the Tate resolution in the degrees $0 \leq a \leq n$. In our example, this complex only has two terms:

```
T=tateResolution(M,low,high);
W = beilinsonWindow T
      6      1
      E  <-- E
betti W
      0 1
total: 6 1
      0: 6 .
      1: . 1
```

The Beilinson monad itself is obtained by applying a certain U functor (see [EES15, §2]) to any piece of $T(\mathcal{F})$ which contains the Beilinson window. This whole process is computed directly from the module M via

```
B = beilinsonMonad M
```

which, in this case, gives a two-term complex:

```
6
S  <-- cokernel {1, 1} | x_(1,2) |
                  {1, 1} | -x_(1,1) |
                  {1, 1} | x_(1,0) |
```

Macaulay2 computes sufficiently many terms of the Tate resolution of \mathcal{F} (as in §2), extracts the subquotient complex corresponding to the Beilinson window, and then applies the functor U .

The 0th-cohomology of the Beilinson monad is a module representing the sheaf \mathcal{F} , while the other cohomology groups are trivial as sheaves—that is, supported on the irrelevant ideal. Note also that the 0th-cohomology may not be equal to the module M , but the two modules will agree after truncating at a sufficiently positive multi-degree. For instance:

```
isIsomorphic(HH^0 B ,M)
false
isIsomorphic (truncate({0,0},HH^0 B),truncate({0,0},M))
true
```

Similarly, the other cohomology of B will have irrelevant support, and therefore will be 0 in sufficiently positive⁴ multi-degrees.

⁴For the precise meaning of a **sufficiently positive degree** see [EES15, Definition 1.8].

5. THE DIRECT IMAGE COMPLEX

This package can also compute the pushforward complex of a sheaf along a morphism to projective space. The main ingredient is the pushforward to a factor in a product of projective spaces, which we explain first.

Let $I \subset \{0, \dots, t-1\}$ be a proper subset and let

$$\pi: \mathbb{P} \rightarrow \mathbb{P}^I := \prod_{i \in I} \mathbb{P}^{n_{i+1}}$$

be the projection to the corresponding factors. (Note that Macaulay2 prefers integers sequences starting with 0 in many commands, which is the reason for the re-indexing.) Given a coherent sheaf \mathcal{F} on \mathbb{P} , represented by an S -module M , the function

`directImageComplex(M, I)`

computes a complex of $S_I = \text{Sym}(\oplus_{i \in I} W_{i+1})$ -modules, whose sheafification is a Beilinson monad for the direct image complex $R\pi_*\mathcal{F}$ in $D^b(\mathbb{P}^I)$. See [EES15, Corollary 0.3] for the related theorem.

The key idea is that an appropriate multi-graded strand of the Tate resolution of \mathcal{F} is a Tate resolution for $R\pi_*(\mathcal{F})$ on the image. (While we have focused on Tate resolutions of sheaves, one can also define the Tate resolution of an element in the derived category, such as $R\pi_*(\mathcal{F})$.)

For example, let $\mathcal{F} = \mathcal{O}_{\mathbb{P}^1 \times \mathbb{P}^2}$, represented by the module S^1 and let $\pi: \mathbb{P}^1 \times \mathbb{P}^2 \rightarrow \mathbb{P}^1$. As in §3, we have:

```
cohomologyMatrix(S^1, {-3,-3}, {3,3})
| 20h 10h 0 10 20 30 40 |
| 12h 6h 0 6 12 18 24 |
| 6h 3h 0 3 6 9 12 |
| 2h h 0 1 2 3 4 |
| 0 0 0 0 0 0 0 |
| 0 0 0 0 0 0 0 |
| 2h3 h3 0 h2 2h2 3h2 4h2 |
```

The far left vertical column of this matrix corresponds to the subquotient of $\mathbf{T}(\mathcal{F})$ consisting of summands generated in degree $(-3, i)$ for some $-3 \leq i \leq 3$. That subcomplex, which only involves exterior variables corresponding to the \mathbb{P}^2 -factor, is a finite part of the Tate resolution of $R\pi_*(\mathcal{F}(-3, 0))$ on \mathbb{P}^2 . It can be computed by

```
T = tateResolution(S^1, {-3,-3}, {3,3});
s = strand(T, {-3,0}, {0});
```

Note that by the Künneth formula, $R^0\pi_*\mathcal{F}(-3, 0) = 0$ and

$$R^1\pi_*\mathcal{F}(-3, 0) \cong H^1(\mathbb{P}^1, \mathcal{O}_{\mathbb{P}^1}(-3)) \otimes \mathcal{O}_{\mathbb{P}^2}[-1] \cong \mathcal{O}_{\mathbb{P}^2}^2[-1]$$

where $[-1]$ indicates that the sheaf lies in cohomological degree 1. We can view the Betti table of this strand:

```

betti s
      -1  0  1  2  3  4  5  6
o76 = total: 20 12 6 2 2 6 12 20
      1: 20 12 6 2 . . . .
      2: . . . . . . . .
      3: . . . . 2 6 12 20

```

Note that, since it is a strand, the terms retain the same homological that they had within the Tate resolution. Combining this with the methods of §4, we can compute a Beilinson monad for $R\pi_*(\mathcal{F}(-3, 0))$ directly from this strand of its Tate resolution.

We now show how this enables one to solve the general problem of computing $R\pi_*\mathcal{F}$ of a sheaf \mathcal{F} on a projective scheme X under a morphism $\pi: X \rightarrow \mathbb{P}^m$. To specify the morphism π , we suppose that X is given by its homogeneous ideal $J \subset R = K[x_0, \dots, x_n]$. Locally near any point of X the morphism can be given by $m + 1$ homogeneous forms of the same degree, but a different representation is necessary at points where all these forms are 0. To represent π at every point of X at once, we thus allow k such representations, which we package as a $k \times (m+1)$ matrix $\phi = (\phi_{ij})$ of homogeneous forms in R , such that any two forms in the same row have the same degree. To ensure that the different rows define the same morphism where both are nonzero, we insist that the 2×2 minors of ϕ are contained in J ; to ensure that the morphism has no base locus, we insist that the entries of ϕ have no common zero on $X = V(J)$.

Represent \mathcal{F} by an R -module N (whose annihilator contains J) and let T denote the homogeneous coordinate ring of \mathbb{P}^m . The function

```
directImageComplex(J, N, phi)
```

returns a complex of T -modules whose sheafification is a Beilinson monad for $R\pi_*\mathcal{F}$.

Example 5.1. Consider the cubic scroll $\mathbb{P}(\mathcal{O}_{\mathbb{P}^1}(2) \oplus \mathcal{O}_{\mathbb{P}^1}(1)) \cong X \subset \mathbb{P}^4$ defined by the ideal of minors of the 2×3 matrix

$$m = \begin{pmatrix} x_0 & x_1 & x_3 \\ x_1 & x_2 & x_4 \end{pmatrix}.$$

Let $\pi: X \rightarrow \mathbb{P}^1$ be defined at each point by the ratio of (at least) one of the columns of m , so that the map π is represented, in the sense above, by the matrix $\phi = \text{transpose } m$.

Let $N = \text{Sym}^2(\text{coker } m) \otimes R(1)$. The sheaf \mathcal{F} associated to N is the line bundle $\mathcal{F} = \mathcal{O}_X(1) \otimes \pi^*\mathcal{O}_{\mathbb{P}^1}(2)$ on X . We compute:

```

kk = ZZ/101;
R = kk[x_0..x_4];
m = matrix{{x_0, x_1, x_3}, {x_1, x_2, x_4}}

```

```

J = minors(2,m);
N = symmetricPower(2,coker m)**R^{1};
phi = transpose m;
RpiN = directImageComplex(J,N,phi);
T = ring RpiN;
RpiN

```

$$\begin{array}{ccc}
& 9 & 7 \\
T & \leftarrow & T \\
0 & & 1
\end{array}$$

We have thus $R\pi_*\mathcal{F}$ is represented by the complex $\mathcal{O}_{\mathbb{P}^1}^9 \leftarrow \mathcal{O}_{\mathbb{P}^1}(-1)^7$. Using this, one can check that $R^1\pi_*\mathcal{F} = 0$ and $R^0\pi_*\mathcal{F} \cong \mathcal{O}_{\mathbb{P}^1}(3) \oplus \mathcal{O}_{\mathbb{P}^1}(4)$.

6. CORNER COMPLEXES.

Let M be a finitely generated, multi-graded module representing a coherent sheaf \mathcal{F} on \mathbb{P} and let c be a multi-degree such that $bgg(M_{\geq c})$ is acyclic. The corner complex is an exact complex of the form

$$\text{Tail}_c(M) \rightarrow bgg(M_{\geq c}).$$

The exactness of the corner complexes, and the fact that $\text{Tail}_c(M)$ equals a subquotient complex of $\mathbf{T}(\mathcal{F})$ in appropriate degrees, are the essential features which allow one to compute Tate resolutions in the first place.

Since corner complexes are so fundamental to this theory, we also introduced a function `cornerComplex` for computing any portion of a corner complex at c , at least assuming that $bgg(M_{\geq c})$ is acyclic. In §2, we described how to compute $\text{Tail}_c(M)$ in this case. By also computing enough terms of the injective resolution $bgg(M_{\geq c})$, which by 1.2 amounts to linear algebra, we can thus compute any portion of a corner complex.

Example 6.1. We again consider $\mathbb{P}^1 \times \mathbb{P}^2$, with homogeneous coordinate ring S . We compute the corner complex of S^1 at $c = \{0, 0\}$:

```

r = cornerComplex(S^1, {0,0}, {-4,-4}, {3,3});
beti r

```

	-6	-5	-4	-3	-2	-1	0	1	2	3	4	5	6	7
total:	40	54	50	35	15	5	1	1	5	15	35	70	126	210
0:	40	54	50	35	15	5	1
1:
2:
3:
4:	1	5	15	35	70	126	210

This computation thus recovers terms corresponding to sheaf cohomology groups H^0 and H^3 . In particular, the corner map goes from $E^1(-2, -3)$ to E^1 . If we were to increase c , then we would recover some H^1 and H^2 groups as well.

For further information on corner complexes, see [EES15, §3] and the documentation for the function `composedFunctions`.

REFERENCES

- [AO89] Vincenzo Ancona and Giorgio Ottaviani, *An introduction to the derived categories and the theorem of Beilinson*, Atti Accad. Peloritana Pericolanti Cl. Sci. Fis. Mat. Natur. **67** (1989), 99–110 (1991).
- [Beĭ78] A. A. Beĭlinson, *Coherent sheaves on \mathbf{P}^n and problems in linear algebra*, Funktsional. Anal. i Prilozhen. **12** (1978), no. 3, 68–69 (Russian). MR509388 (80c:14010b)
- [BGG78] I. N. Bernšteĭn, I. M. Gelfand, and S. I. Gelfand, *Algebraic vector bundles on \mathbf{P}^n and problems of linear algebra*, Funktsional. Anal. i Prilozhen. **12** (1978), no. 3, 66–67 (Russian). MR509387 (80c:14010a)
- [EES15] David Eisenbud, Daniel Erman, and Frank-Olaf Schreyer, *Tate resolutions for products of projective spaces*, Acta Math. Vietnam. **40** (2015), no. 1, 5–36.
- [EFS03] David Eisenbud, Gunnar Fløystad, and Frank-Olaf Schreyer, *Sheaf cohomology and free resolutions over exterior algebras*, Trans. Amer. Math. Soc. **355** (2003), no. 11, 4397–4426 (electronic).
- [Huy06] D. Huybrechts, *Fourier-Mukai transforms in algebraic geometry*, Oxford Mathematical Monographs, The Clarendon Press, Oxford University Press, Oxford, 2006.
- [M2] Daniel R. Grayson and Michael E. Stillman, *Macaulay2, a software system for research in algebraic geometry*. Available at <http://www.math.uiuc.edu/Macaulay2/>.
- [M2-BGG] Hirotachi Abo, Wolfram Decker, David Eisenbud, Frank-Olaf Schreyer, Gregory G. Smith, and Michael E. Stillman, *BGG, package for Macaulay2*. Available at <http://www.math.uiuc.edu/Macaulay2/>.
- [M2-Tate] David Eisenbud, Daniel Erman, and Frank-Olaf Schreyer, *TateOnProducts, package for Macaulay2*. Available at <http://www.math.uiuc.edu/Macaulay2/>.

DEPARTMENT OF MATHEMATICS, UNIVERSITY OF CALIFORNIA AT BERKELEY AND
THE MATHEMATICAL SCIENCES RESEARCH INSTITUTE, BERKELEY, CA 94720, USA
E-mail address: de@msri.org

DEPARTMENT OF MATHEMATICS, UNIVERSITY OF WISCONSIN, MADISON, WIS-
CONSIN, 53706, USA
E-mail address: derman@math.wisc.edu

FACHBEREICH MATHEMATIK, UNIVERSITÄT DES SAARLANDES, CAMPUS E2 4,
D-66123 SAARBRÜCKEN, GERMANY
E-mail address: schreyer@math.uni-sb.de